

Membuat Add In Terbilang Untuk Microsoft Excel

Budi Hartono

pakne.dhea@gmail.com

http://cakbud.info

Lisensi Dokumen:

Copyright © 2003-2006 IlmuKomputer.Com

Seluruh dokumen di IlmuKomputer.Com dapat digunakan, dimodifikasi dan disebarkan secara bebas untuk tujuan bukan komersial (nonprofit), dengan syarat tidak menghapus atau merubah atribut penulis dan pernyataan copyright yang disertakan dalam setiap dokumen. Tidak diperbolehkan melakukan penulisan ulang, kecuali mendapatkan ijin terlebih dahulu dari IlmuKomputer.Com.

Artikel berikut akan membahas bagaimana membuat membuat Add In terbilang (menterjemahkan angka menjadi kata-kata) untuk Microsoft Excel. Bagi anda yang sering membuat tanda terima pembayaran tentunya sering kali anda harus menterjemahkan angka-angka menjadi kata-kata, misal **Rp. 1000,-** diterjemahkan menjadi **Seribu Rupiah**, tentu akan sangat melelahkan jika hal ini harus dikerjakan secara manual, dimana anda harus mengeja setiap angka pada transaksi anda untuk kemudian anda tuliskan.

Dengan Add In ini saya berharap akan bisa membantu pekerjaan anda dalam membuat tanda terima. Untuk mempersingkat proses, saya tidak menjelaskan apa itu Add In, anda bisa pelajari sendiri dari Microsoft Excel help.

Persiapan

Inti dari artikel ini adalah membuat Add Ins terbilang bukan membuat fungsi terbilang, sehingga persiapan pertama yang perlu anda lakukan adalah membuat fungsi terbilang menggunakan VBA, kalau anda merasa kesulitan, anda bisa googling dengan kunci pencarian “fungsi terbilang VBA”, dijamin dapat banyak J . Kalau anda tidak bisa googling karena keburu pulang dari warnet atau koneksi internetnya sudah terlanjur disconnect, anda bisa juga menggunakan kode fungsi terbilang dibawah ini, dengan catatan anda tidak boleh menyebutkan bahwa kode fungsi ini adalah karya penulis, karena penulis sendiri tidak pernah merasa membuat kode tersebut J , penulis mendapatkan kode ini sekitar tahun 2000 an dengan bahasa xBase (DBASEIII+), yang kemudian penulis modifikasi dengan VBA agar bisa jalan di Microsoft Access, Microsoft Excel dan Microsoft Visual Basic, tapi penulis sudah lupa dari situs mana.

Berikut Kode Untuk Fungsi Terbilang :

```
Public Function terbilang(x As Currency)
Dim triliun As Currency
Dim milyar As Currency
Dim juta As Currency
Dim ribu As Currency
Dim satu As Currency
```

```

Dim sen As Currency
Dim baca As String

'Jika x adalah 0, maka dibaca sebagai 0
If x = 0 Then
    baca = angka(0, 1)
Else
    'Pisah masing-masing bagian untuk triliun, milyar, juta, ribu, rupiah, dan
    sen
    triliun = Int(x * 0.001 ^ 4)
    milyar = Int((x - triliun * 1000 ^ 4) * 0.001 ^ 3)
    juta = Int((x - triliun * 1000 ^ 4 - milyar * 1000 ^ 3) / 1000 ^ 2)
    ribu = Int((x - triliun * 1000 ^ 4 - milyar * 1000 ^ 3 - juta * 1000 ^ 2) /
    1000)
    satu = Int(x - triliun * 1000 ^ 4 - milyar * 1000 ^ 3 - juta * 1000 ^ 2 -
    ribu * 1000)
    sen = Int((x - Int(x)) * 100)
    'Baca bagian triliun dan ditambah akhiran triliun
    If triliun > 0 Then
        baca = ratus(triliun, 5) + "triliun "
    End If
    'Baca bagian milyar dan ditambah akhiran milyar
    If milyar > 0 Then
        baca = ratus(milyar, 4) + "milyar "
    End If
    'Baca bagian juta dan ditambah akhiran juta
    If juta > 0 Then
        baca = baca + ratus(juta, 3) + "juta "
    End If
    'Baca bagian ribu dan ditambah akhiran ribu
    If ribu > 0 Then
        baca = baca + ratus(ribu, 2) + "ribu "
    End If
    'Baca bagian satuan
    If satu > 0 Then
        baca = baca + ratus(satu, 1) + "rupiah "
    else
        baca = baca + "rupiah "
    End If
    'Baca bagian sen dan ditambah akhiran sen
    If sen > 0 Then
        baca = baca + ratus(sen, 0) + "sen"
    End If
End If
terbilang = UCase(Left(baca, 1)) & LCase(Mid(baca, 2))
End Function

Function ratus(x As Currency, posisi As Integer) As String
Dim a100 As Integer, a10 As Integer, a1 As Integer
Dim baca As String
a100 = Int(x * 0.01)
a10 = Int((x - a100 * 100) * 0.1)
a1 = Int(x - a100 * 100 - a10 * 10)
'Baca Bagian Ratus
If a100 = 1 Then
    baca = "Seratus "
Else
    If a100 > 0 Then
        baca = angka(a100, 2) + "ratus "
    End If
End If
'Baca Bagian Puluh dan Satuan
If a10 = 1 Then
    baca = baca + angka(a10 * 10 + a1, 2)
Else

```

```
If a10 > 0 Then
    baca = baca + angka(a10, 2) + "puluh "
End If
If a1 > 0 Then
    If posisi = 2 And a100 = 0 And a10 = 0 Then
        baca = baca + angka(a1, 1)
    Else
        baca = baca + angka(a1, 2)
    End If
End If
End If
ratus = baca
End Function


Function angka(x As Integer, posisi As Integer)
Select Case x
    Case 0: angka = "Nol"
    Case 1:
        If posisi = 2 Then
            angka = "Satu "
        Else
            angka = "Se"
        End If
    Case 2: angka = "Dua "
    Case 3: angka = "Tiga "
    Case 4: angka = "Empat "
    Case 5: angka = "Lima "
    Case 6: angka = "Enam "
    Case 7: angka = "Tujuh "
    Case 8: angka = "Delapan "
    Case 9: angka = "Sembilan "
    Case 10: angka = "Sepuluh "
    Case 11: angka = "Sebelas "
    Case 12: angka = "Dua belas "
    Case 13: angka = "Tiga belas "
    Case 14: angka = "Empat belas "
    Case 15: angka = "Lima belas "
    Case 16: angka = "Enam belas "
    Case 17: angka = "Tujuh belas "
    Case 18: angka = "Delapan belas "
    Case 19: angka = "Sembilan belas "
End Select
End Function
```

Wew... panjang banget, pasti banyak diantara pembaca yang beranggapan kodenya terlalu bertele-tele, tidak simpel atau tidak rekursif, penulis pun sering merasa demikian dan tugas anda untuk menyederhanakan dan jangan lupa kalau sudah sederhana kirim kodenya ke penulis J .
Sekarang mari kita mulai.


Membuat Module

Langkah pertama yang perlu dilakukan adalah membuat modul, caranya :

1. Buka Microsoft Excel Anda
2. Klik Pada Menu Tools – Macro – Visual Basic Editor (Alt+F11)




3. Setelah tampil jendela Visual Basic Editor, langkah selanjutnya adalah menambahkan module baru dengan cara Klik menu Insert – Module



Dari Proses ini akan terbentuk satu modul baru dengan nama Module1.

4. Masukkan fungsi terbilang yang sudah anda siapkan kedalam module1



5. Sampai disini proses sudah setengah jadi, pindahkan jendela aktif lembar kerja excel, coba anda tuliskan pada cell B2 dengan **1000** dan pada cell B3 tuliskan dengan **=terbilang(B2)**, apa yang terjadi?, kalau pada cell B3 tampil **Seribu**

rupiah, anda boleh melanjutkan ke langkah berikutnya, tapi kalau tampil #NAME,#VALUE atau yang lain selain **Seribu rupiah** anda harus perbaiki fungsi terbilang anda pada Module1 terlebih dahulu.



Sampai disini proses pembuatan module telah selesai, kita lanjutkan ke langkah berikutnya.

Membuat deskripsi dan mengamankan kode anda

Langkah selanjutnya adalah memberi deskripsi dan mengamankan kode VBA anda, langkahnya adalah sebagai berikut:

1. Pada lembar kerja Excel anda, klik menu Tools – Macros
2. Ketikkan nama fungsi terbilang anda dan pastikan tombol Option menjadi enable (menyala), kalau tidak, bisa dipastikan anda salah menuliskan nama fungsi terbilangnya.




3. Klik Option, dan ketikkan Deskripsi fungsi terbilang anda dalam kotak Description. Akhiri proses dengan klik tombol **OK** dan berikutnya Klik Tombol **Cancel** untuk menutup Macro dialog.
4. Untuk memberi password kode VBA anda, pindahkan jendela aktif ke jendela Visual Basic, Klik kanan pada VBAProject (Book1) – pilih VBAProject Properties, selanjutnya akan ditampilkan dialog VBAProject Properties , aktifkan Tab protection, masukkan password pada kotak password dan konfirmasi pada kotak Confirm Password.
Jika anda bermaksud mendistribusikan Add Ins anda secara gratis dan untuk tujuan pembelajaran, sebaiknya tidak perlu dipassword, karena penerima Add Ins anda akan lebih mudah mempelajari kode anda.

Membuat Add Ins

Setelah semua langkah kita lakukan, selanjutnya adalah proses packing Add In, adapun langkahnya adalah sebagai berikut.

1. Tutup Visual basic Editor anda. Simpan Dokumen excel anda dengan klik menu **File – Save As**.
2. Ubah **Save As Type** menjadi **Microsoft Excel Add In (*.xla)**, dan spesifikasikan tempat penyimpanannya.
3. Beri nama bebas, misal **budi-terbilang.xla** dan tutup dokumen excel anda boleh disimpan boleh tidak.




Pembuatan Add In telah selesai, dan anda bisa mendistribusikan file **budi-terbilang.xla** untuk teman anda atau siapa saja yang membutuhkan.
Apakah pekerjaan sudah selesai? Belum, kita lanjutkan ke pekerjaan terakhir.

Menginstall Add In

Setelah kita membuat, selanjutnya adalah menginstall Add In yang sudah kita buat pada Microsoft Excel kita atau teman anda. Langkahnya sebagai berikut:

1. Siapkan File Add In (Budi-terbilang.xla)
2. Buka Microsoft Excel anda, Klik Menu Tools – Add Ins
3. Klik Browse dan arahkan ke tempat penyimpanan file Add In Anda (budi-terbilang.xla) dan klik OK
4. Pastikan budi-terbilang tampil dalam list dan dalam kondisi aktif (✓)



5. Nah... pekerjaan telah selesai, dan sekarang anda bisa menggunakan fungsi terbilang di cell manapun dan file manapun (selama Add In aktif tentunya J)

C	D	E	F	G	H	I
300	Tiga ratus	rupiah				
7.000	Tujuh ribu	rupiah				
56.000	Lima puluh enam ribu	rupiah				
3.450.000	Tiga juta empat ratus lima puluh ribu	rupiah				
233.300.400	Dua ratus tiga puluh tiga juta tiga ratus ribu empat ratus	rupiah				
5500	Lima ribu	lima ratus	rupiah			

Demikian semoga membantu dan bermanfaat, tentunya tulisan ini jauh dari sempurna, karenanya saran dan masukan selalu penulis tunggu dari pembaca semua.

Ujicoba dilakukan dengan :
OS – Microsoft Windows Xp
Microsoft Excel 2002

Referensi:

1. Build An Excel Add In, Martin Green (www.fontstuff.com)
Thanks For Your Inspiring Articles

Biografi Penulis



Budi Hartono. Sampai saat ini masih terus belajar untuk menjadi bagian dari solusi. Menulis berbagai artikel di situs <http://cakbud.info>.
Bisa dihubungi di pakne.dhea@gmail.com